Процедура GetServiceCost()

ПараметрыЗаказов = ПолучитьПараметрыСообщения();

// Создадим Заказы DPD.

WSПрокси = WSСсылки.DPDGetServiceCost.СоздатьWSПрокси(

"http://dpd.ru/ws/calculator/2012-03-20",

"DPDCalculatorService",

"DPDCalculatorPort"

);

ФабрикаСервиса = WSПрокси.ФабрикаXDTO;

Для Каждого Параметры Из ПараметрыЗаказов.ОбщиеПараметры Цикл 

// serviceCostRequest

ЭлементServiceCostRequest = ФабрикаСервиса.Создать(

ФабрикаСервиса.Тип("http://dpd.ru/ws/calculator/2012-03-20", "serviceCostRequest")

);

// auth

ЭлементAuth = ФабрикаСервиса.Создать(

ФабрикаСервиса.Тип("http://dpd.ru/ws/calculator/2012-03-20", "auth")

);

ЭлементAuth.clientNumber = Число(clientNumber);

ЭлементAuth.clientKey    = clientKey;

ЭлементServiceCostRequest.auth = ЭлементAuth;

// cityRequest

ЭлементCityRequest = ФабрикаСервиса.Создать(

ФабрикаСервиса.Тип("http://dpd.ru/ws/calculator/2012-03-20", "cityRequest")

);

КонтактыОтправителя = ПараметрыЗаказов.КонтактыОтправителя.НайтиСтроки(Новый Структура(

"order", Параметры.order)

);

Если КонтактыОтправителя.Количество() Тогда 

ЭлементCityRequest.cityId      = Число(КонтактыОтправителя[0].cityId);

ЭлементCityRequest.index       = КонтактыОтправителя[0].postcode;

ЭлементCityRequest.cityName    = КонтактыОтправителя[0].cityName;

ЭлементCityRequest.regionCode  = Строка(КонтактыОтправителя[0].regionCode);

ЭлементCityRequest.countryCode = КонтактыОтправителя[0].countryCode;

ЭлементServiceCostRequest.pickup = ЭлементCityRequest;

КонецЕсли;

// cityRequest

ЭлементCityRequest = ФабрикаСервиса.Создать(

ФабрикаСервиса.Тип("http://dpd.ru/ws/calculator/2012-03-20", "cityRequest")

);

КонтактыПолучателя = ПараметрыЗаказов.КонтактыПолучателя.НайтиСтроки(Новый Структура(

"order", Параметры.order)

);

Если КонтактыПолучателя.Количество() Тогда 

ЭлементCityRequest.cityId      = Число(КонтактыПолучателя[0].cityId);

ЭлементCityRequest.index       = КонтактыПолучателя[0].postcode;

ЭлементCityRequest.cityName    = КонтактыПолучателя[0].cityName;

ЭлементCityRequest.regionCode  = Строка(КонтактыПолучателя[0].regionCode);

ЭлементCityRequest.countryCode = КонтактыПолучателя[0].countryCode;

ЭлементServiceCostRequest.delivery = ЭлементCityRequest;

КонецЕсли;

ЭлементServiceCostRequest.selfPickup    = Ложь;

ЭлементServiceCostRequest.selfDelivery  = Ложь;

ЭлементServiceCostRequest.weight        = Параметры.cargoWeight;

ЭлементServiceCostRequest.volume        = Параметры.cargoVolume;

ЭлементServiceCostRequest.serviceCode   = Параметры.serviceCode;

ЭлементServiceCostRequest.pickupDate    = datePickup;

ЭлементServiceCostRequest.declaredValue = Параметры.cargoValue;

// getServiceCost

ЭлементGetServiceCost = ФабрикаСервиса.Создать(

ФабрикаСервиса.Тип("http://dpd.ru/ws/calculator/2012-03-20", "getServiceCost")

);

ЭлементGetServiceCost.request = ЭлементServiceCostRequest;

Результат = ОтправитьСообщение(

"getServiceCost",

"http://ws.dpd.ru/services/calculator2?wsdl",

ФабрикаСервиса,

ЭлементGetServiceCost,

"request"

);

Если Не Результат = Неопределено И Результат.Статус = 200 Тогда 

НоваяСтрока = КодыУслуг.Добавить();

НоваяСтрока.order       = Параметры.order;

НоваяСтрока.serviceCode = ПолучитьСписокУслуг(Результат.Ответ);

КонецЕсли;

КонецЦикла;

КонецПроцедуры // GetServiceCost()

Процедура CreateOrder()

ПараметрыЗаказов = ПолучитьПараметрыСообщения();

// Создадим Заказы DPD.

WSПрокси = WSСсылки.DPDCreateOrder.СоздатьWSПрокси(

"http://dpd.ru/ws/order2/2012-04-04",

"DPDOrderService",

"DPDOrderPort"

);

ФабрикаСервиса = WSПрокси.ФабрикаXDTO;

Отправители = ПараметрыЗаказов.ОбщиеПараметры.Скопировать(, "Организация");

Отправители.Свернуть("Организация");

Для Каждого ТекущийОтправитель Из Отправители Цикл 

ЗаказыОтправителя = ПараметрыЗаказов.ОбщиеПараметры.НайтиСтроки(

Новый Структура("Пометка, Организация", Истина, ТекущийОтправитель.Организация)

);

Если Не ЗаказыОтправителя.Количество() Тогда 

Продолжить;

КонецЕсли;

ОбщиеПараметры = ЗаказыОтправителя[0];

// dpdOrdersData

ЭлементDpdOrdersData = ФабрикаСервиса.Создать(

ФабрикаСервиса.Тип("http://dpd.ru/ws/order2/2012-04-04", "dpdOrdersData")

);

// auth

ЭлементAuth = ФабрикаСервиса.Создать(

ФабрикаСервиса.Тип("http://dpd.ru/ws/order2/2012-04-04", "auth")

);

ЭлементAuth.clientNumber = Число(clientNumber);

ЭлементAuth.clientKey    = clientKey;

ЭлементDpdOrdersData.auth = ЭлементAuth;

// Информация, содержащаяся в этом параметре, является общей для всех заказов в запросе.

ЭлементHeader = ФабрикаСервиса.Создать(

ФабрикаСервиса.Тип("http://dpd.ru/ws/order2/2012-04-04", "header")

);

ЭлементHeader.datePickup       = datePickup;

ЭлементHeader.pickupTimePeriod = pickupTimePeriod;

ЭлементHeader.payer            = Число(clientNumber);

ЭлементHeader.regularNum       = "";

ЭлементHeader.senderAddress    = ПолучитьОписаниеАдреса(

ФабрикаСервиса,

ПараметрыЗаказов.КонтактыОтправителя.Найти(ОбщиеПараметры.order, "order")

);

ЭлементDpdOrdersData.header = ЭлементHeader;

Для Каждого ТекущийЗаказ Из ЗаказыОтправителя Цикл 

ЭлементOrder = ФабрикаСервиса.Создать(

ФабрикаСервиса.Тип("http://dpd.ru/ws/order2/2012-04-04", "order")

);

ЗаполнитьЗначенияСвойств(ЭлементOrder, ТекущийЗаказ);

ЭлементOrder.receiverAddress = ПолучитьОписаниеАдреса(

ФабрикаСервиса,

ПараметрыЗаказов.КонтактыПолучателя.Найти(ТекущийЗаказ.order, "order")

);

ЭлементDpdOrdersData.order.Добавить(ЭлементOrder);

КонецЦикла;

ЭлементCreateOrder = ФабрикаСервиса.Создать(

ФабрикаСервиса.Тип("http://dpd.ru/ws/order2/2012-04-04", "createOrder")

);

ЭлементCreateOrder.orders = ЭлементDpdOrdersData;

Результат = ОтправитьСообщение(

"createOrder",

?(ТестовыйРежим, "http://wstest.dpd.ru/services/order2?wsdl", "http://ws.dpd.ru/services/order2?wsdl"),

ФабрикаСервиса,

ЭлементCreateOrder,

"orders"

);

Если Не Результат = Неопределено И Результат.Статус = 200 Тогда 

ЗарегистрироватьСтатусыЗаявок(Результат.Ответ);

КонецЕсли;

КонецЦикла;

КонецПроцедуры // CreateOrder()

Функция ПрочитатьОтвет(Ответ)

ЧтениеXML = Новый ЧтениеXML;

ЧтениеXML.УстановитьСтроку(Ответ);

ПостроительDOM = Новый ПостроительDOM;

Возврат ПостроительDOM.Прочитать(ЧтениеXML);

КонецФункции // ПрочитатьОтвет()

Функция ПолучитьСписокУслуг(Ответ)


СписокУслуг = Новый СписокЗначений;


ДокументDOM = ПрочитатьОтвет(Ответ);


ЭлементыReturn = ДокументDOM.ПолучитьЭлементыПоИмени("return");


Для Каждого Услуга Из ЭлементыReturn Цикл 



ПредставлениеУслуги = "";



serviceCode         = Неопределено;



serviceName         = Неопределено;



ЭлементServiceName = Услуга.ПолучитьЭлементыПоИмени("serviceName");



Если ЭлементServiceName.Количество() Тогда 




serviceName = ЭлементServiceName[0].ТекстовоеСодержимое;




ПредставлениеУслуги = serviceName;



КонецЕсли;



ЭлементCost = Услуга.ПолучитьЭлементыПоИмени("cost");



Если ЭлементCost.Количество() Тогда 




ПредставлениеУслуги =





ПредставлениеУслуги +





?(ПустаяСтрока(ПредставлениеУслуги), "", "; стоимость: ") +





ЭлементCost[0].ТекстовоеСодержимое;




//



КонецЕсли;



ЭлементDays = Услуга.ПолучитьЭлементыПоИмени("days");



Если ЭлементDays.Количество() Тогда 




ПредставлениеУслуги =





ПредставлениеУслуги +





?(ПустаяСтрока(ПредставлениеУслуги), "", "; срок: ") +





ЭлементDays[0].ТекстовоеСодержимое + " дн.";




//



КонецЕсли;



ЭлементServiceCode = Услуга.ПолучитьЭлементыПоИмени("serviceCode");



Если ЭлементServiceCode.Количество() Тогда 




serviceCode = ЭлементServiceCode[0].ТекстовоеСодержимое;




СписокУслуг.Добавить(serviceCode, ПредставлениеУслуги);



КонецЕсли;


КонецЦикла;


Возврат СписокУслуг;

КонецФункции // ПолучитьСписокУслуг()

Процедура ЗарегистрироватьСтатусыЗаявок(Ответ)


ДокументDOM = ПрочитатьОтвет(Ответ);


СвойстваЗаявок.Очистить();


ЭлементыReturn = ДокументDOM.ПолучитьЭлементыПоИмени("return");


Для Каждого Заказ Из ЭлементыReturn Цикл 



ЭлементOrderNumberInternal = Заказ.ПолучитьЭлементыПоИмени("orderNumberInternal");



Если Не ЭлементOrderNumberInternal.Количество() Тогда 




Продолжить;



КонецЕсли;



НоваяСтрока = СвойстваЗаявок.Добавить();



НоваяСтрока.orderNumberInternal = ЭлементOrderNumberInternal[0].ТекстовоеСодержимое;



ЭлементOrderNum = Заказ.ПолучитьЭлементыПоИмени("orderNum");



Если ЭлементOrderNum.Количество() Тогда 




НоваяСтрока.orderNum = ЭлементOrderNum[0].ТекстовоеСодержимое;



КонецЕсли;



ЭлементStatus = Заказ.ПолучитьЭлементыПоИмени("status");



Если ЭлементStatus.Количество() Тогда 




НоваяСтрока.status = ЭлементStatus[0].ТекстовоеСодержимое;



КонецЕсли;



ЭлементErrorMessage = Заказ.ПолучитьЭлементыПоИмени("errorMessage");



Если ЭлементErrorMessage.Количество() Тогда 




НоваяСтрока.errorMessage = ЭлементErrorMessage[0].ТекстовоеСодержимое;



КонецЕсли;


КонецЦикла;


// Далее, описание кода по принятому алгоритму регистрации результатов создания Заявок...

КонецПроцедуры // ЗарегистрироватьСтатусыЗаявок()
